Motion in a central field

The objective of this document is to investigate the motion of a single point through a central field, with a specific goal of showing that planetary orbits are conic sections.

Kepler's laws

Newton's derivation of Kepler's laws was a watershed event in the history of science and mathematics. Kepler, largely based on data meticulously gathered by Tycho Brahe, wrote down a precise description of planetary motion.

Kepler's description takes the form of the following three laws (as stated by Wikipedia):

1. The orbit of every planet is an ellipse, with the Sun at one of the foci of the ellipse.
2. A line joining a planet and the Sun sweeps out equal areas during equal intervals of time.
3. The square of the orbital period of a planet is directly proportional to the cube of the semi-major axis of the ellipse.

It should be emphasized that Kepler's techniques are purely inductive, being based on Brahe's empirical data. Newton's contribution was purely deductive, that is he derived Kepler's laws from more fundamental laws, namely his second law of motion and his law of universal gravitation. Newton's laws are applicable to many more contexts, however.

Gravitation

As we've discussed, the force on a planet moving through a solar system with a massive star can be modeled by $\boldsymbol{F}(\boldsymbol{x}) = -gm\boldsymbol{x} \big/ \|\boldsymbol{x}\|^3$. Since Newton's second law states that $\boldsymbol{F} = m\boldsymbol{x}''$, we get $\boldsymbol{x}'' = -g\boldsymbol{x} \big/ \|\boldsymbol{x}\|^3$. This is a vector equation that should be considered as three dimensional for our purposes. Nonetheless, we will assume for the moment that the motion is planar and write $\boldsymbol{x} = \langle x_1, x_2 \rangle$. (We can prove that the motion is planar anyway, after we learn a bit about angular momentum.) Thus, we can now write our second order equation as a two-dimensional system:

$$\begin{array}{rcl} x_1'' & = & -\frac{gx_1}{\left(x_1^2+x_2^2\right)^{3/2}}, \\ x_2'' & = & -\frac{gx_2}{\left(x_1^2+x_2^2\right)^{3/2}} \end{array} . \tag{1}$$

Given a specific value of g (we will typically take $g = 1$) and initial conditions, we can solve this system numerically. Here's how to do so with *Mathematica*, assuming the initial conditions $x_1(0) = 1$, $x_2(0) = 0$, $x_1'(0) = 0$, $x_2'(0) = 0.7$.

```
Clear[x1, x2];
{x1[t_], x2[t_]} = {x1[t], x2[t]} /. First[NDSolve[{
      x1''[t] == -x1[t] / (x1[t]^2 + x2[t]^2)^(3/2),
      x2''[t] == -x2[t] / (x1[t]^2 + x2[t]^2)^(3/2),
      x1[0] == 1, x1'[0] == 0, x2[0] == 0, x2'[0] == 0.7},
    {x1[t], x2[t]}, {t, 0, 5}]];
T = t /. FindRoot[x2[t] == 0, {t, 3.5}];
path = ParametricPlot[{x1[t], x2[t]}, {t, 0, T}]
```



Certainly looks elliptical with the origin at a focus of the ellipse. Ultimately, our objective is to prove this fact.

Conservative fields

Definition and examples

An n-dimensional vector field $\boldsymbol{F}$ is called *conservative* if it arises as the gradient of a function $f : \mathbb{R}^n \to \mathbb{R}$, i.e. $\boldsymbol{F} = \nabla f$ for some f. The function $V = -f$ is called the *potential* or *potential energy* of $\boldsymbol{F}$.

As an example, the field $\boldsymbol{F}_1(x, y) = \langle 2xy^3, 3x^2y^2 \rangle$ is conservative, since it can be expressed as the gradient of $f(x, y) = x^2y^3$. The field $\boldsymbol{F}_2(x, y) = \langle xy^3, x^2y^2 \rangle$ is not conservative; try to find a real valued function f such that $\nabla f = \boldsymbol{F}_2$.

Natural questions arise. How can we tell if a given vector field is conservative? How would we know if $\boldsymbol{F}_1$ above is conservative, if we didn't have $f(x, y) = x^2y^3$? How do we know for sure that $\boldsymbol{F}_2$ above is not conservative? Perhaps, we just weren't clever enough to find the appropriate f. If we suspect that a vector field $\boldsymbol{F}$ is conservative, how do we find the function f such that $\nabla f = \boldsymbol{F}$?

Suppose we write $\boldsymbol{F}(x, y) = \langle u(x, y), v(x, y) \rangle$. In our first example $\boldsymbol{F}_1$ above, we have $u(x, y) = 2xy^3$

and $v(x, y) = 3x^2y^2$. Note that $u_y = v_x = 6xy^2$. This is not by chance, for Clairaut's theorem asserts that the mixed partial derivatives of a function are equal - i.e. $f_{xy} = f_{yx}$. This provides a test to check if a given vector field is conservative.

We can now see that the field $\boldsymbol{F}_2$ is definitely not conservative. If it were, say $\boldsymbol{F}_2 = \nabla f$, then we would have

$$3xy^2 = \frac{\partial}{\partial y}\left(xy^3\right) = f_{xy} = f_{yx} = \frac{\partial}{\partial x}\left(x^2y^2\right) = 2xy^2.$$

Exercise 1: Which of the following fields are conservative?

(a) $\boldsymbol{F}(x, y) = \langle y, x\rangle$

(b) $\boldsymbol{F}(x, y) = \langle x, y\rangle$

(c) $\boldsymbol{F}(x, y) = \langle y\cos(xy), x\cos(xy)\rangle$

(d) $\boldsymbol{F}(x, y) = \langle x\cos(xy), y\cos(xy)\rangle$

Exercise 2: Prove that the gravitational field that arises from equations 1 is conservative by showing they arise from the potential function $f(x_1, x_2) = -g \Big/ \sqrt{x_1^2 + x_2^2}$.

Finding the potential

As we know, the field $\boldsymbol{F}(x, y) = \left\langle 2xy^3, 3x^2y^2\right\rangle$ is conservative, since it has potential function $f(x, y) = x^2y^3$. What if we don't know f? Well, we do know that $f_x = 2xy^3$. Integrating, we obtain $f(x, y) = x^2y^3 + c(y)$. The notation $c(y)$ indicates that the "constant" of integration is in general a function of y. Now, if f has this form, then $f_y = 3x^2y^2 + c'(y)$. On the other hand, the potential function of $\boldsymbol{F}$ must satisfy $f_y = 3x^2y^2$. Thus, $c'(y) = 0$ so $c(y)$ is indeed a constant in this case. Any constant will do but the simplest is probably zero.

As another example, consider $\boldsymbol{F}(x, y) = \left\langle 2xy^2 + 1, 2x^2y + 1\right\rangle$. If f is to be a potential function for $\boldsymbol{F}$, then $f_x = 2xy^2 + 1$. Thus, $f(x, y) = x^2y^2 + x + c(y)$ and $f_y(x, y) = 2x^2y + c'(y)$. This implies that $c'(y) = 1$ so $c(y) = y$. Finally, we obtain $f(x, y) = x^2y^2 + x + y$.

Exercise 3: Find the potentials for the conservative fields from exercise 1.

Conservation of energy

There is a law of conservation of energy for conservative systems that mirrors what we have learned for conservative equations. Since the equation is assumed to be conservative, we can write it in the form $\boldsymbol{F} = \nabla f$. If we wish to emphasize the state vector $\boldsymbol{x}$, we could write it in the form $\boldsymbol{F}(\boldsymbol{x}) = \nabla f(\boldsymbol{x})$. In this context, the *kinetic energy* is $\|\boldsymbol{x}'\|/2$, the *potential energy* is $V(\boldsymbol{x}) = -f(\boldsymbol{x})$, and the total energy is the sum of these two. A fundamental fact is that the total energy is conserved.

Theorem 1 *In a conservative system, to total energy is conserved. In symbols,*

$$E = \frac{1}{2}\|\boldsymbol{x}'\|^2 + V(\boldsymbol{x})$$

is constant.

Proof: We'll simply show that $dE/dt = 0$. Using the multivariate chain rule on $V(\boldsymbol{x})$, we get $dV/dt = \nabla V \cdot \boldsymbol{x}'$. Applying the dot product rule to $\|\boldsymbol{x}'\|^2 = \boldsymbol{x}' \cdot \boldsymbol{x}'$ we get

$$\frac{d\,\|\boldsymbol{x}'\|^2}{\mathrm{dt}} = \frac{d\,(\boldsymbol{x}' \cdot \boldsymbol{x}')}{dt} = \boldsymbol{x}'' \cdot \boldsymbol{x}' + \boldsymbol{x}' \cdot \boldsymbol{x}'' = 2\boldsymbol{x}' \cdot \boldsymbol{x}''.$$

Thus

$$\frac{dE}{dt} = \frac{d}{dt}\left(\frac{1}{2}\,\|\boldsymbol{x}'\|^2 + V(\boldsymbol{x})\right) = \boldsymbol{x}' \cdot \boldsymbol{x}'' + \nabla V \cdot \boldsymbol{x}' = \boldsymbol{x}' \cdot (\boldsymbol{x}'' + \nabla V) = \boldsymbol{x}' \cdot (\boldsymbol{x}'' - \nabla f) = 0.\square$$

A major difference between conservative *equations* versus conservative *systems* is that *every* equation of the form of the form $x'' = F(x)$ is conservative, since we can always integrate F to find the potential function. By contrast, a given system $\boldsymbol{x}'' = \boldsymbol{F}(\boldsymbol{x})$ might or might not be conservative, since $\boldsymbol{F}$ might or might not arise as the gradient of a potential function.

Exercise 4: Use the technique in the proof of theorem 1 to show that total energy is conserved for a conservative equation.

Exercise 5: Suppose that we add the initial conditions $x_1(0) = 2$, $x_2(0) = 0$, $x_1'(0) = 0$, and $x_2'(0) = 1$ to the gravitational system given by equations 1. Find the total energy in the system.

Polar form and angular momentum

At this point, we've generalized the notion of energy to the point that it applies to systems. If we have a 2^{nd} order, two dimensional system, say

$$\begin{aligned} x_1'' &= f(x_1, x_2) \\ x_2'' &= g(x_1, x_2) \end{aligned}$$

then we might introduce variables $y_1 = x_1'$ and $y_2 = x_2'$ to obtain the first order, four dimensional system

$$\begin{aligned} y_1' &= f(x_1, x_2) \\ y_2' &= g(x_1, x_2) \\ x_1' &= y_1 \\ x_2' &= y_2 \end{aligned}$$

Thus the phase space is four-dimensional. The law of conservation of energy places one constraint on the variables x_1, x_2, y_1, and y_2. Thus the solution of the first order system is forced to flow along a three dimensional subset of this four dimensional space. (Such a set is called a manifold.) If we ultimately want to prove that the orbits of the gravitational system are elliptical, then we'll need to reduce this further. The next major step is to introduce a second conserved quantity called the angular momentum, a concept which is best described in polar form.

Polar form

We'll write the polar form of a parametrized path in the plane as $\boldsymbol{r}(t) = r(t)\boldsymbol{e}_r(t)$, or just $\boldsymbol{r} = r\boldsymbol{e}_r$. The scalar r is the distance of the point to the origin and the vector $\boldsymbol{e}_r$ is a unit vector pointing in the direction of the object. With this notation, we can express the equation of motion in a central field as $\boldsymbol{r}'' = \Phi(r)\boldsymbol{e}_r$.

It should be emphasized that this version of the equations of motion makes no specific reference to the dimension of the space. (That's why I've been using the ambiguous term "polar form" in favor of the precise term "polar coordinates".) We will prove, however, that motion in a 3D central field necessarily lies in a plane. Thus, it is typically safe to think of polar form as polar coordinates. In this context, we have $\boldsymbol{e}_r(t) = \langle \cos(\theta(t)), \sin(\theta(t)) \rangle$.

Conservation of angular momentum

The *angular momentum* of the motion parametrized by $\boldsymbol{r}$ is defined to be $\boldsymbol{M} = \boldsymbol{r} \times \boldsymbol{r}'$. Note that we are implicitly assuming that the system is three dimensional. If the motion is planar, we embed it in three space in the natural way to allow the formation of the cross-product. Either way, the angular momentum is a vector quantity. The main fact concerning angular momentum is reminiscent of the corresponding fact for energy.

Theorem 2 *If $\boldsymbol{r}$ denotes the position of an object in space moving through a central field, then the angular momentum of $\boldsymbol{r}$ is conserved.*

Proof: We simply take the derivative of the angular momentum.

$$\frac{d\boldsymbol{M}}{dt} = \frac{d}{dt}\left(\boldsymbol{r} \times \boldsymbol{r}'\right) = \boldsymbol{r}' \times \boldsymbol{r}' + \boldsymbol{r} \times \boldsymbol{r}'' = \boldsymbol{0}.$$

Note that both terms in the sum are $\boldsymbol{0}$ (the zero vector), since the vectors in each product are parallel.□

Recall that the cross-product of two vectors is perpendicular to both. This fact, combined with theorem 2, implies that the motion of a point through a central field must lie in a plane. This is illustrated in figure 1.

As an example, let's compute the angular momentum of the system defined in equations 1 back in the gravitation section. Since angular momentum is conserved, we can use the initial conditions $x_1(0) = 1$, $x_2(0) = 0$, $x_1'(0) = 0$, $x_2'(0) = 0.7$ to determine it. Embedding the x_1x_2-plane into 3-space, we obtain the position and velocity vectors $\langle 1, 0, 0 \rangle$ and $\langle 0, 0.7, 0 \rangle$. The cross product of these two vectors $\langle 0, 0, 0.7 \rangle$. This is the angular momentum of the system.

Kepler's second law

Kepler's second law states that a line joining a planet and the Sun sweeps out equal areas during equal intervals of time. This is illustrated in figure 2. Ultimately, this is a simple consequence of the conservation of angular momentum.

To prove Kepler's second law, we introduce the unit vector $\boldsymbol{e}_\theta$ perpendicular to $\boldsymbol{e}_r$ and in the direction of rotation. We can then express the velocity $\boldsymbol{r}'$ in terms of the basis $\boldsymbol{e}_r$, $\boldsymbol{e}_\theta$, as in lemma 1.



Figure 1: The angular momentum vector



Figure 2: Equal area swept out in equal time

Lemma 1

$$\boldsymbol{r}' = r'\boldsymbol{e}_r + r\theta'\boldsymbol{e}_\theta.$$

Proof: The vector $\boldsymbol{e}_r$ rotates with constant angular velocity θ'. (This is the definition of angular velocity.) Since $\boldsymbol{e}_\theta$ maintains a constant right angle to $\boldsymbol{e}_r$, it rotates with the same angular velocity. Since the derivative of a unit vector is perpendicular to that unit vector, we have $\boldsymbol{e}_r' = \theta'\boldsymbol{e}_\theta$ and $\boldsymbol{e}_\theta' = -\theta'\boldsymbol{e}_r$. Thus, differentiating $\boldsymbol{r} = r\boldsymbol{e}_r$, we get

$$\boldsymbol{r}' = r'\boldsymbol{e}_r + r\boldsymbol{e}_r' = r'\boldsymbol{e}_r + r\theta'\boldsymbol{e}_\theta.\Box$$

As a consequence of lemma 1, we can express the angular momentum as follows.

$$\boldsymbol{M} = \boldsymbol{r} \times \boldsymbol{r}' = \boldsymbol{r} \times (r'\boldsymbol{e}_r + r\theta'\boldsymbol{e}_\theta) = \boldsymbol{r} \times r'\boldsymbol{e}_r + r\theta'\boldsymbol{r} \times \boldsymbol{e}_\theta = r^2\theta'\boldsymbol{e}_r \times \boldsymbol{e}_\theta.$$

In particular, $r^2\theta'$ represents the length of the angular momentum. Since $\boldsymbol{M}$ is preserved, its length must be constant so $r^2\theta'$ is constant. From this fact, Kepler's second law can be derived fairly easily since it's clear that in figure 2 we have

$$\Delta\mathrm{A} = A(t + \Delta\mathrm{t}) - A(t) \approx \frac{1}{2}r^2\frac{d\theta}{dt}\Delta\mathrm{t}.$$

Reduction of dimension

We continue to focus on our polar description of motion in a central field described by $\boldsymbol{r}'' = \boldsymbol{F}(\boldsymbol{r})$, where $\boldsymbol{F} = -\nabla U$. Since the field is central, U should depend only upon the magnitude of $\boldsymbol{r}$. Thus, it should cause no ambiguity to think of U as a function of one variable and write $U(r)$.

Theorem 3 *For motion in a central field, the distance from the center of the field varies in the same way as r varies in the one-dimensional problem with potential energy*

$$V(r) = U(r) + \frac{M^2}{2r^2}.$$

$V(r)$ is called the *effective potential* of the system. It should be emphasized that r and M are scalar quantities in the one-dimensional problem. They are related to the vectors $\boldsymbol{r}$ and $\boldsymbol{M}$ in the multidimensional problem via $r = \|\boldsymbol{r}\|$ and $M = \|\boldsymbol{M}\|$.

Before jumping into a proof of this theorem, it might be good to look at an example. Consider the system defined in equations 1 defined back in the section on gravitation. Assuming that the code from that section has been executed, we can plot the distance of the object to the origin as a function of time as follows.

```
pic2 = Plot[Norm[{x1[t], x2[t]}], {t, 0, T},
  AxesOrigin → {0, 0}]
```



We now try to relate this to the one dimensional problem with potential $V(r) = U(r) + M^2/\left(2r^2\right)$. Using the fact that the force function $F(r) = -V'(r)$, we can find the force function (the right hand side) for the one-dimensional problem. Thus the one-dimensional problem is

$$r'' = \frac{M^2}{r^3} - U'(r).$$

Back in exercise 2 you should have shown that the gravitational field has potential function

$$f\left(x_1, x_2\right) = -\frac{g}{\sqrt{x_1^2 + x_2^2}}.$$

Setting $r = \sqrt{x_1^2 + x_2^2}$, namely the distance of the point to the origin, we obtain $U(r) = -g/r$. Of course, we set $g = 1$ for our example, so this simplifies further to $U(r) = -1/r$. Now, we've already computed the angular momentum of our system is 0.7. Thus the effective potential can be written as

$$V(r) = -\frac{1}{r} + \frac{0.49}{2r^2}$$

and the one-dimensional problem is

$$r'' = \frac{0.49}{r^3} - \frac{1}{r^2}.$$

We can solve this problem numerically, plot it, and compare the result to distance versus time graph obtained from the system. Here's the result.

```
Clear[r];
r[t_] = r[t] /. First[NDSolve[{r''[t] == 0.49/r[t]^3 - 1/r[t]^2,
    r[0] == 1, r'[0] == 0}, r[t], {t, 0, T}]];
pic1 = Plot[r[t], {t, 0, T}, AxesOrigin → {0, 0}]
```



Looks the same!

Elliptical orbits

Using the equation of conservation of energy in the one-dimensional problem, it's fairly easy to deduce the dependence of r upon t. We have

$$\frac{1}{2}(r')^2 + V(r) = E.$$

Thus,

$$\frac{dr}{dt} = \sqrt{2(E - V(r))}.$$

Now, we applying the chain rule to $d\theta/dt$, we get $\frac{d\theta}{dt} = \frac{d\theta}{dr}\frac{dr}{dt}$. Furthermore, $d\theta/dt = M/r^2$ by conservation of angular momentum. Thus,

$$\frac{M}{r^2} = \frac{d\theta}{dr}\sqrt{2(E - V(r)}.$$

This is a separable equation, so

$$\theta = \int d\theta = \int \frac{M}{r^2\sqrt{2(E - V(r))}} dr.$$

If we can evaluate the integral on the right and solve for r, then we've got a polar description of the orbit. In the case of planetary motion, we have

$$V(r) = \frac{M^2}{2r^2} - \frac{g}{r}$$

Thus,

$$\theta = \int \frac{M}{r^2\sqrt{2\left(E - M^2/\left(2r^2\right) + g/r\right)}}\,dr = \arccos\left(\frac{M/r - g/M}{\sqrt{2E + g^2/M^2}}\right). \tag{2}$$

Here's "proof" that the derivative of the arccos term yields the integrand.

```
FullSimplify[
 D[ArcCos[(M/r - g/M) / Sqrt[2 E + g^2 / M^2]], r] -
  M / (r^2 Sqrt[2 (E - M^2 / (2 r^2) + g / r)]),
 Assumptions → {M ≥ 0, r ≥ 0, g ≥ 0}]
```

```
0
```

If we set

$$\begin{aligned} p &= M^2/g \\ e &= \sqrt{1 + 2EM^2/g^2} \end{aligned}, \tag{3}$$

then $\theta = \arccos((p/r) - 1)$, i.e. $r = p/(1 + e\cos(\theta))$. This is the polar form of a conic section.

Let's make the correspondence between this polar equation and the gravitational system defined in equations 1 quite explicit. Back in the gravitation section, we used the initial conditions $x_1(0) = 1$, $x_2(0) = 0$, $x_1'(0) = 0$, $x_2'(0) = 0.7$. We've computed the angular momentum to be $M = 0.7$ and the energy to be $E = \frac{1}{2}0.7^2 - 1 = -0.755$. Recall also that we've also assumed $g = 1$. We can use these to try to regenerate the orbit obtained before like so:

```
p = 0.7^2;
e = Sqrt[1 + 2 (-0.755) 0.7^2];
polarPath[θ_] := p / (1 + e * Cos[θ]);
PolarPlot[polarPath[θ], {θ, 0, 2 Pi}]
```



If you compare this path to the path back in the gravitational section, it looks to have the exact same dimensions, but it's been flipped. Let's check the dimensions:

```
{{x1[0], polarPath[π]}, {x1[T / 2], polarPath[0]}}
```

```
{{1., 1.}, {-0.324503, 0.324503}}
```

Looks good, but why's it been flipped? If we examine equation 2, we note that the integral could include an arbitrary constant. Setting this constant to π, we generate the exact same path as back in the gravitational section.

```
PolarPlot[polarPath[θ + Pi], {θ, 0, 2 Pi}]
```



Exercise 6: The polar form of a conic section is

$$r(\theta) = \frac{p}{1 + e\cos(\theta)}.$$

The parameter $e \geq 0$ is called the *eccentricity* of the conic. $0 \leq e < 1$ yields an ellipse, with $e = 0$ a circle. $e = 1$ yields a parabola and $e > 1$ yields a hyperbola. Use equations 3 to find initial conditions for the gravitational equations 1 that yield exactly a parabola. Use Mathematica to draw the parabola twice - once using `ParametricPlot` applied to the result from `NDSolve` and once using `PolarPlot` applied to the polar equation.

Other central fields

It's important to realize that many of the results in this document apply to motion through *any* central field - not just the gravitational field. The justification of Kepler's second law, for example, made no specific reference to the gravitational field and holds for any central field. The proof that planetary orbits are conic sections, on the other hand, made quite specific reference to the gravitational field and does not hold more generally. In this section, we'll play with some other central fields.

Here's a crazy example. Suppose we want to write down a system that has potential energy $U(r) = r$. (What could be simpler?) In terms of x and y, this means that $U(x, y) = \sqrt{x^2 + y^2}$. Computing minus the gradient, this yields the system

$$\begin{aligned} x'' &= -x \Big/ \sqrt{x^2 + y^2} \\ y'' &= -y \Big/ \sqrt{x^2 + y^2} \end{aligned}$$

Let's choose initial conditions $x(0) = 1$, $x'(0) = 0$, $y(0) = 0$, $y(0) = 1/2$ and investigate the orbit numerically.

```
Clear[x, y];
r[t_] = {x[t], y[t]} /. First[NDSolve[{
        x''[t] == -x[t] / (x[t]^2 + y[t]^2)^(1/2),
        y''[t] == -y[t] / (x[t]^2 + y[t]^2)^(1/2),
        x[0] == 1, x'[0] == 0,
        y[0] == 0, y'[0] == 1/2},
      {x[t], y[t]}, {t, 0, 400}]];
ParametricPlot[r[t], {t, 0, 400}]
```



Glad I don't live in that universe!

There's actually a fairly simple explanation of what's going on. Recall that the effective potential is

$$V(r) = U(r) + \frac{M^2}{2r^2} = r + \frac{1}{8r^2},$$

since $M = 1/2$. Now, the total energy can be computed from the initial conditions to be 9/8. Since the kinetic energy is positive, the radius must remain the region where $V(r) < 9/8$. Let's plot this.

```
Plot[{r + 1/(8 r^2), 9/8}, {r, 0, 3},
 AspectRatio → Automatic,
 AxesOrigin → {0, 0}]
```



We can compute the points of intersection fairly precisely.

```
NSolve[r + 1/(8 r^2) == 1 + 1/8, r]
```

```
{{r → 1.}, {r → 0.421535}, {r → -0.296535}}
```

Thus the orbit bops back and forth from $r = 1$ to $r \approx 0.42$. It never quite returns back to the starting point so the interior of an annulus is traced out. This behavior is actually quite typical. In fact, in the *only* potential functions that always lead to closed orbits have the form $U(r) = -g/r$ (gravity) and $U(r) = ar^2$! I wonder what $U(r) = ar^2$ leads to?

Exercise 7: Write down the system corresponding to the potential function $U(r) = r^2$. Solve the system explicitly using the initial conditions $x(0) = 1$, $x'(0) = 0$, $y(0) = 0$, $y'(0) = 1/2$.

Exercise 8: Write down the system corresponding to the potential function $U(r) = \sin(r^2)$. Investigate it numerically.

The two-body problem

Suppose the universe is a plane containing precisely two point masses that interact via gravitation. (We consider planar motion for ease of visualization; however, planar motion frequently occurs naturally and the ideas generalize readily to motion in space.) Now an object of mass m_1 at location $\boldsymbol{r}_1 = \langle x_1, y_1 \rangle$ exerts a force on an object of mass m_2 at location $\boldsymbol{r}_2 = \langle x_2, y_2 \rangle$ according to Newton's inverse square law:

$$\boldsymbol{F} = Gm_1m_2 \frac{\boldsymbol{r}_1 - \boldsymbol{r}_2}{|\boldsymbol{r}_1 - \boldsymbol{r}_2|^3},$$

where G is a fundamental constant. By Newton's 2^{nd} law, this induces an acceleration on object 2: $m_2\boldsymbol{r}_2''$. This yields a vector differential equation describing the motion of object 2:

$$m_2\boldsymbol{r}_2'' = Gm_1m_2 \frac{\boldsymbol{r}_1 - \boldsymbol{r}_2}{|\boldsymbol{r}_1 - \boldsymbol{r}_2|^3},$$

which is equivalent to the following *pair* of equations for the x and y components:

$$m_2x_2'' = Gm_1m_2 \frac{x_1 - x_2}{\left((x_1 - x_2)^2 + (y_1 - y_2)^2\right)^{3/2}}$$

$$m_2y_2'' = Gm_1m_2 \frac{y_1 - y_2}{\left((x_1 - x_2)^2 + (y_1 - y_2)^2\right)^{3/2}}.$$

Of course, $\boldsymbol{r}_1 = \langle x_1, y_1 \rangle$ also moves according to very similar equations.

$$m_1x_1'' = Gm_1m_2 \frac{x_2 - x_1}{\left((x_1 - x_2)^2 + (y_1 - y_2)^2\right)^{3/2}}$$

$$m_1y_1'' = Gm_1m_2 \frac{y_2 - y_1}{\left((x_1 - x_2)^2 + (y_1 - y_2)^2\right)^{3/2}}.$$

As it turns out, there are conservation laws of energy, angular momentum, and momentum for this system. (These conservation laws can be formulated even for the mutual interaction of n points.) In fact, the two body problem can ultimately be written in the form

$$m_1\boldsymbol{r}_1'' = -\nabla_1 U\left(\|\boldsymbol{r}_1 - \boldsymbol{r}_2\|\right),$$

$$m_2\boldsymbol{r}_2'' = -\nabla_2 U\left(\|\boldsymbol{r}_1 - \boldsymbol{r}_2\|\right),$$

where ∇_1 denotes the gradient with respect to $\boldsymbol{r}_1$ and ∇_2 denotes the gradient with respect to $\boldsymbol{r}_2$. From these facts, one can prove the following theorem.

Theorem 4 *The time variation of $\boldsymbol{r} = \boldsymbol{r}_2 - \boldsymbol{r}_1$ in the two body problem is the same as that for the motion of a point of mass $m = m_1m_2/\left(m_1 + m_2\right)$ in a field with potential $U(\|\boldsymbol{r}\|)$.*



Figure 3: Motion in the two body problem.

As a result, the points move along a conic section with one focus at their center of mass, in the case of an inverse square law. This is illustrated in figure 3.

These ideas can be extended to n-bodies but at $n = 3$ there are chaotic solutions with no analytic description. Thus, we really need to focus on numerical techniques and the right place to do that is in a *Mathematica* notebook.

To do

Proof that all central fields are conservative. Proof of theorem 3.